
Nutzung der Soundkanäle

PLAY Tongenerierung mittels Notennamen bzw. -nummern

PLAY(n) Abfrage der Musikpuffer

SOUND Direkte Programmierung des Soundgenerators

Mit dem im Computergrundgerät vorhandenen Soundgenerator können für die
einzelnen Tonkanäle (max. 3 Kanäle 1)) beliebige Töne bzw. Rauschen (nur
mit Sound) in den verschiedensten Frequenzen erzeugt werden. Der Tonumfang
beträgt 8 Oktaven. Weiterhin gibt es fest vorgegebene Hüllkurven, deren
Abarbeitungsgeschwindigkeit veränderlich ist.
Nach dem Netzeinschalten, mit dem RBASIC-Befehl BEEP und bei jeder Fehler-
meldung wird der Soundgenerator mit Standardparametern initialisiert.
Am Ende dieses Abschnittes folgt eine zusammenfassende Darstellung der
Anweisungen PLAY und SOUND.

Tongenerierung mittels Notennamen

Format: PLAY zka(,zkb(,zkc))

zkn - Zeichenkette (Konstante oder Variable) für den jeweiligen
Soundkanal, die den zu spielenden Tonverlauf beinhaltet.
Die Zuordnung der Zeichenketten zu den Tonkanälen erfolgt
entsprechend ihrer Position in der PLAY-Anweisung, d.h. zka
wird dem Kanal A, usw. zugewiesen.

Funktion: Ein zu spielender Tonverlauf wird mit Hilfe spezieller Musik-
Macro-Kommandos beschrieben. Auf diese Weise können u.a. die
Noten direkt in zkn stehen. Der Leistungsumfang der PLAY-Anwei-
sung entspricht dem der DRAW-Anweisung für grafische Ausgaben.

Musik-Makro-Kommandos

An, Bn, Cn, Dn, En, Fn, Gn Der erste Buchstabe stellt die Notenbezeich-
A#n,..,G#n bzw.A+n,..,G+n nung dar (der deutschen Note H entspricht
A-n,...,G-n (n=1...64) dabei die internationale Bezeichnung B). Die

(Standard: n=4) Halbtöne werden durch die nachgestellten
Zeichen # oder + (z.B. Fis mit F# oder F+)
bzw. durch das Zeichen - (z.B. Des mit D- )
gekennzeichnet. Ein # oder + bzw. ein -
bedeutet, daß ein Halbton höher bzw. tiefer
gespielt wird.
Der Parameter n gibt die Länge der Note in
1/n-tel an (1/1, 1/2, 1/4,..,1/64). Es können
auch Zwischenwerte (z.B. n=3) angegeben werden.
Ist n=0 bzw. nicht angegeben, so wird
automatisch die Länge angenommen, die durch
den Befehl L definiert wurde.

Beispiel 1: 10 BEEP :REM Standardwerte einstellen
20 PLAY "C4 D8 E8 F2 G4 A8 B8 C2":REM Tonleiter mit unter-

schiedlich langen Noten

1) z.Zt. (bis Ende 1990) ist nur Kanal A im vollen Umfang nutzbar.



Ln (n=1..64) Einstellen der Länge (in 1/n-tel) für die
(Standard: n=4) Noten, denen kein Längenparameter folgt. Die

Einstellung kann für jeden Kanal getrennt
erfolgen. Wird n nicht angegeben, so wird n=4
als länge eingestellt.

Beispiel 2: 10 BEEP :REM Standardwerte einstellen
15 PLAY "L8" :REM Notenlänge für Kanal A
20 PLAY "C4 D E F2 G4 A B C2" :REM Tonleiter wie im Beispiel 1

Nn(;l) (n=1..95, l=1..64) Festlegen der zu spielenden Note anhand ihrer
Position innerhalb der chromatischen
Tonleiter, d.h. vom tiefsten (n=1) bis zum
höchsten Ton (n=95) mit allen Halbtönen. Die
Länge (in 1/l-tel) der entsprechenden Note
kann durch Angabe eines 2. Parameters (l)
eingestellt werden. Ist kein Längenparameter
angegeben, so wird die für den jeweiligen
Kanal eingestellte Notenlänge benutzt.

Beispiel 3: Im Unterschied zu den vorangegangenen Beispielen wird die letzte
Note der Tonleiter in der 5. Oktave gespielt (PLAY "N48"
entspricht PLAY "O5 C").

10 BEEP :REM Standardwerte einstellen
20 PLAY "L8" :REM Notenlänge für Kanal A
30 PLAY "N36;4 N38 N40 N41;2 N43;4 N45 N47 N48;2"

On (n=1..8) Mit diesem Macro-Kommando kann die momentan
(Standard: n=4) eingestellte Oktave (nach Netzeinschalten bzw.

BEEP gilt n=4) verändert werden. Die
Einstellung der Oktave kann für jeden Kanal
gesondert erfolgen und behält ihre Gültigkeit
bis zum nächsten O-Kommando. Wird n nicht
angegeben, so gilt n=4. Eine Oktave beginnt
immer bei C und endet bei B (H).

Beispiel 4: Dreistimmige Tonausgabe, wobei Kanal A in der 4. und Kanal C in
der 5. Oktave spielt. Im Kanal B wird dagegen nur die Note C in
der 5. Oktave gespielt.

10 BEEP
20 PLAY "L2","L2","L2 O5" :REM Notenlänge vereinbaren
30 PLAY "GGGF+G","B O5 C O4 B A B","DCDDD"

Rn (n=1..64) Definiert eine Pause bestimmter Länge in der
(Standard: n=4) Form 1/n. Sinnvoll sind z.B. 1/1, 1/2, 1/4,

..,1/64 Pausen. Zwischenwerte sind ebenfalls
möglich. Wird n nicht angegeben, so gilt
ebenfalls der Standardwert (n=4).

Beispiel 5: Im Gegensatz zu der Tonleiter im Beispiel 3 ist im folgenden
Beispiel eine deutliche Pause (R4) zwischen den Noten F und G
hörbar.

10 BEEP :REM Standardwerte einstellen
20 PLAY "L8" :REM Notenlänge für Kanal A
30 PLAY "C4 D E F4 R4 G4 A B O5 C2"

Mn (n=1..65535) Festlegen der Periodendauer für die mit dem
(Standard: n=255) Befehl S festgelegte Hüllkurve. Je kleiner n

ist, desto schneller wird die charakteristi
sche Wellenform (siehe S) durchlaufen.



Beispiel 6: In diesem Beispiel wird die gleiche Tonfolge wie im Beispiel 5
ausgegeben, allerdings wird der Lautstärkeverlauf durch eine
Hüllkurve (in Zeile 25) bestimmt.

10 BEEP :REM Standardwerte einstellen
20 PLAY "L8" :REM Notenlänge für Kanal A
25 PLAY "S1 M1800" :REM max. Lautstärke, abschwellend
30 PLAY "C4 D E F4 R4 G4 A B O5 C2"

Sn (n=1..15) Festlegen des Lautstärkeverlaufes einer Ton
folge mittels 8 verschiedener Hüllkurven. Der
Lautstärkeverlauf wird somit durch die jewei
lige Hüllkurve gesteuert, d.h. eine zuvor mit
Vn programmierte Lautstärke ist wirkungslos.
Einige Parameter ergeben die gleiche Hüll
kurve. Bei fehlendem Parameter gilt n=1.
Folgende Hüllkurven sind möglich:

Parameter n ! Kurve ! Lautstärkeverlauf
-----------------------------------------
0,1,2,3,9 ! @_____ ! sofort max. Lautstärke, dann abklingend und aus
4,5,6,7,15 ! /_____ ! ansteigend bis max. Lautstärke, dann aus
8 ! @@@@@@ ! anfangs ganz laut, dann leiser werdend und

! ! danach wieder ganz laut usw.
10 ! @/@/@/ ! anfangs ganz laut, dann abklingend und wieder

! ! langsam ansteigend usw.
11 ! @ ! anfangs ganz laut, dann leiser werdend und

! ! danach sofort max. Lautstärke, so bleibend
12 ! ////// ! ansteigend bis max. Lautstärke, dann aus usw.
13 ! / ! ansteigend auf max. Lautstärke, so bleibend
14 ! /@/@/@ ! langsam ansteigend, danach abklingend usw.

Beispiel 7: Dreistimmige Tonausgabe mit jeweils aktiver Hüllkurve. In
diesem Fall können alle Kanäle nur mit der gleichen Hüllkurve
arbeiten. (Beachten Sie bitte die weiter unten angegebenen
Hinweise zu diesem Sachverhalt !)

10 BEEP
15 PLAY "S1 M1800","S1 M1800","S1 M1800 O5":REM Hüllkurve 1
20 PLAY "GGGF+G","B O5 C O4 BAB","DEDDD" :REM alles 1/4-Noten

Tn (n=32..255) Festlegen der Tongeschwindigkeit. Der Para
(Standard: n=120) meter n gibt die Anzahl der zu spielenden

1/4-Noten pro Minute an. Damit können
Musikstücke sehr variabel im Tempo gestaltet
werden. Bei fehlendem Parameter gilt n=120.
Das Kommando ist ebenfalls für die Kanäle
getrennt verwendbar.

Beispiel 8: Durch Einfügen der Zeile 28 in das Programm von Beispiel 6
ergibt sich das folgende Beispiel. Hier kann durch Variation
des Parameters von T (in Zeile 28) dessen Wirkung erprobt
werden.

10 BEEP
20 PLAY "L8" :REM Notenlänge für Kanal A
25 PLAY "S1 M1800" :REM max. Lautstärke, abschwellend
28 PLAY "T200" :REM 200 1/4-Noten pro Minute
30 PLAY "C4 D E F4 R4 G4 A B O5 C2"

Vn (n=0..15) Einstellen der Lautstärke für die einzelnen
(Standard: n=8) Kanäle. Eine zuvor aktivierte Hüllkurve wird



mit dem Befehl V für den betreffenden Kanal
wirkungslos.
Bei n=0 wird für den jeweiligen Kanal die
Tonausgabe beendet. Fehlt der Parameter, so
wird n=8 verwendet.

Beispiel 9: Wird die Zeile 25 vom Beispiel 8 durch die Zeile

25 PLAY "S1 M1800 V4" :REM Hüllkurve wirkungslos !!!

ersetzt, so wird mit V4 eine konstante Lautstärke eingestellt,
d.h. die die zuvor vereinbarte Hüllkurve ist wirkungslos.

. (Punkt) Durch Anfügen eines Punktes an eine Note bzw.
Pause wird die Note bzw. Pause um die Hälfte
des ursprünglichen Wertes verlängert.

X string; Führt die in string enthaltenen Macro
Kommandos aus.

Beispiel 10: Die zu spielende Tonfolge wird einer Zeichenkettenvariable
zugewiesen. die ]bergabe der Variable an die PLAY-Anweisung
erfolgt in der geforderten Form.

10 BEEP: A$="C E G" :REM Tonfolge für Kanal A und B
20 PLAY "O4","O5" :REM Kanal B in der 5. Oktave
30 PLAY "XA$;","XA$;" :REM zweistimmige Tonausgabe

=num-variable; Erlaubt die Verwendung einer numerischen
Variablen zur Angabe der Parameter (n) bei
den zuvor angeführten Macro-Kommandos.
(Numerische Ausdrücke sind nicht erlaubt,
führen bei Verwendung jedoch zu keiner
Fehlermeldung !)

Beispiel 11: 10 BEEP
20 FOR I=36 TO 48
30 PLAY "N=I;" :REM Ausgabe der chromatischen Tonleiter
40 NEXT : in der 4. Oktave

Hinweise: 1. Falls bei mehrstimmiger Tonausgabe für mehrere Kanäle Hüll
kurven programmiert werden, kommt nur die Hüllkurve des
höchstwertigen, in Abarbeitung befindlichen Kanals zur An
wendung. Als höchstwertiger gilt der jeweils am weitesten
rechts stehende Kanal innerhalb der PLAY-Anweisung.

2. Wird in einer PLAY-Anweisung für einen Kanal sowohl das Macro
Kommando V als auch S benutzt, so wirkt nur das in der
Anweisung zuletzt angegebene Kommando.

Abfrage der Musikpuffer

Da die Dauer der Musikausgabe abhängig von den zu spielenden Noten, Pausen,
usw. ist, kann es vorkommen, daß die Abarbeitung der PLAY-Anweisung (durch
den BASIC-Interpreter) eher fertig ist als die eigentliche Tonausgabe. Das
liegt daran, daß die drei Kanäle Puffer haben, in denen durch den Inter-
preter die Macro-Kommandos und Parameter in geeigneter Form abgelegt werden.
Das Lesen dieser Puffer und die Übergabe der Daten an den Soundgenerator
erfolgt interruptgesteuert, d.h. unabhängig vom weiteren Programm. In
manchen Fällen kann es nun wichtig sein, daß man feststellen kann, ob in
den Puffern noch Daten für die Abarbeitung vorhanden sind oder nicht. Diese
Abfrage kann mit der Funktion PLAY(n) erfolgen.



Format: PLAY(n)

n - Nummer des Tonkanals

Funktion: Ist das Ergebnis dieser Funktion gleich -1 (TRUE), so bedeutet
das, daß für den angegebenen Kanal (bei n=1,2,3) bzw. bei einem
der drei Kanäle (n=0) noch Daten im Musikpuffer vorhanden sind.
Eine 0 (FALSE) gibt an, daß keine Daten mehr vorhanden sind.

Beispiel 12: In diesem Beispiel erfolgen die Ausschriften 'Oktave ...'
fortlaufend ohne direkten Bezug zur tatsächlichen Tonausgabe !
Durch Löschen der Anweisung REM in Zeile 80 kommt die Funktion
PLAY(1) zur Anwendung. Bei erneutem Start erfolgen die
Ausschriften synchron mit der Tonausgabe!

10 BEEP
20 FOR I=2 TO 5
30 PRINT "Oktave Nr. ";I
40 J=I*12 : X1=J : X2=J+4 : X3=J+7
50 PLAY "N=X1;" :REM Note C
60 PLAY "N=X2;" :REM Note E
70 PLAY "N=X3;" :REM Note G
80 REM IF PLAY(1)=-1 THEN 80
90 NEXT

Direkte Programmierung des Soundgenerators

Format: SOUND reg,par

reg - Register des Soundgenerators (0...13)
par - Parameter für das selektierte Register

Bedeutung der einzelnen Register und der möglichen Parameter:

Registerpaar 0-1: Mit diesem Registerpaar erfolgt die Einstellung des
par = 0...255 Frequenzwertes (FW) des Tones, welcher im Kanal A
(für Reg. 0) ausgegeben werden soll

Dabei enthält das Register 0 das Low-Byte (FL) des
par = 0...15 Frequenzwertes und das Register 1 das High-Byte (FH) des
(für Reg. 1) im Kanal A auszugebenden Frequenzwertes.

Registerpaar 2-3: wie Registerpaar 0-1, aber für Kanal B

Registerpaar 4-5: wie Registerpaar 0-1, aber für Kanal C

Register 6: Mit dem Wert dieses Registers wird die Rauschfrequenz
par = 0...31 eingestellt, d.h. mit diesem Parameter wird die höchst

mögliche Frequenz der im Frequenzgemisch enthaltenen
Grundwellen festgelegt. Mit par=0 wird die höchste
Frequenz eingestellt. Dieses Register wird für alle 3
Kanäle verwendet, d.h. für die mit Register 7 fest
gelegten Kanäle gilt die gleiche Rauschfrequenz.

Register 7: Mit diesem Register werden die Kanäle ausgewählt, für die
par = 0...31 eine Tonausgabe und/oder das Rauschen erfolgen soll.

Ist das entsprechende Bit von par gesetzt, so ist die
Betriebsart für diesen Kanal aktiv, andernfalls inaktiv.
Es können auch beide Betriebsarten für einen Kanal
gleichzeitig aktiv sein. Für die einzelnen Bits von par
gilt folgende Zuordnung:

Bit n = 0/1 -> Tonausgabe bzw. Rauschen aus/an

n=0,1,2 -> Tonausgabe für Kanal A, B, C
n=3,4,5 -> Rauschen -#-



Register 8,9,10: Diese Register enthalten die Lautstärkewerte für die
par = 0...16 Kanäle A, B, bzw. C. Mit par=0 wird für den jeweiligen

Kanal jegliche Tonausgabe abgeschalten. Mit par=1-15 wird
die Lautstärke eingestellt. Bei par=16 wird die Hüllkurve
aktiviert, d.h. die Lautstärke wird entsprechend der
eingestellten Hüllkurve (mit Register 13) und der Hüll
kurvenfrequenz (Register 11/12) verändert.

Registerpaar 11-12: Mit diesem Registerpaar erfolgt die Einstellung der
par=0...255 Hüllkurvenfrequenz, wobei Register 11 dem Low-Byte (FLH)
(für beide Reg.) und Register 12 dem High-Byte (FHH) entspricht.

Je kleiner der Frequenzwert (FHH*256+FHL) ist, desto
schneller wird die Hüllkurve durchlaufen.

Register 13: Der Wert dieses Registers gibt an, nach welcher Hüllkurve
die Lautstärke verändert wird. Die Zuordnung der mögli
chen Hüllkurven zu den Parametern ist dieselbe wie bei
der PLAY-Anweisung. Damit eine Hüllkurve wirksam wird,
muß diese mit dem Wert 16 im zugehörigen Lautstärke
register aktiviert werden. Der Wert dieses Registers gilt
einheitlich für alle Kanäle (analog zu Register 6).

Hinweise: 1. Für die Tonausgabe muß für den jeweiligen Kanal
- im Register 7 das Bit für die Tonausgabe gleich 1 sein,
- im Lautstärkeregister ein Wert von 1-15 stehen, sowie
- ein geeigneter Frequenzwert (FW) in den Tonregistern stehen.

2. Damit für einen beliebigen Kanal Rauschen erzeugt wird, muß
- im Register 6 eine Rauschfrequenz eingestellt werden,
- im Register 7 das Bit für Rauschen gleich 1 sein, sowie
- im Lautstärkeregister ein Wert von 1-15 stehen.

3. Falls Rauschen und/oder Tonausgabe mit aktiver Hüllkurve
mehrstimmig programmiert wird, so arbeiten die beteiligten
Kanäle mit der gleichen Rauschfrequenz und/oder der gleichen
Hüllkurve.

4. Wenn die Register des Soundgenerators programmiert sind,
erfolgt die Tonausgabe bis zur erneuten Programmierung dieser
Register ständig. D.h. während der Rechner andere Dinge
bearbeitet, kann weiterhin eine Tonausgabe erfolgen.

Beispiel: Mit dem folgenden Beispiel wird das Geräusch eines Hubschraubers
angedeutet. Durch Drücken einer Taste wird die Tonausgabe beendet

10 SOUND 0,111 :REM Frequenzwert (Low-Byte)
20 SOUND 1,0 : für Kanal A (High-Byte)
30 SOUND 6,30 :REM Rauschart einstellen
40 SOUND 8,16 :REM Hüllkurve für Kanal A an
50 SOUND 11,44 :REM Hüllkurvenfrequenz (Low-Byte)
60 SOUND 12,1 : (High-Byte)
70 SOUND 13,14 :REM Hüllkurvenform (auf- und abschwellend)
80 SOUND 7,&B00001001 :REM Kanal A Tonausgabe u. Rauschen an
90 IF INKEY$ =""THEN 90

100 BEEP :REM Standartwerte einstellen



Umrechnungen zwischen der echten Tonfrequenz und dem Frequenzwert

F echte Tonfrequenz in Hz
FT Taktfrequenz des Rechners (3750000) in Hz
FW Frequenzwert
FL Low-Byte des Frequenzwertes
FH High-Byte " "

Tonfrequenz in Frequenzwert

FW = FT / (32 * F) Es gilt: 28 Hz < F < 117188 Hz
FL = FW MOD 256
FH = FW / 256

Frequenzwert in Tonfrequenz

FW = FL + FH * 256 Es gilt: 0 < FL < 255
F = FT / (32 * FW) 0 < FH < 15

Kurzübersicht zu PLAY und SOUND

PLAY-Kommando ! Funktion
-----------------!---------------------------------------------------
C D E F G A B ! Tonname mit möglicher Längenangabe (1...64) und

! Tonhöhenveränderung um 1/2 Ton (+ # -)
O ! Auswahl zwischen 8 Oktaven (1...8)
N ! Tonhöhe durch Zahlenangabe (1...96)
L ! Tonlänge (1...64)
. ! Tonlänge um die Hälfte verlängern
R ! Pause (1...64)
T ! Tempo (32...255)
V ! Lautstärke (0...15)
S ! Hüllkurve (1,4,8,10,11,12,13,14)
M ! Hüllkurvenabstimmung (1...65535)

Sound- ! Funktion ! Bit
register ! ! 7 ! 6 ! 5 ! 4 ! 3 ! 2 ! 1 ! 0
-----------------------------------------------------------------
0 ! Tonfrequenz Kanal A ! FL(A)
1 ! ! x ! x ! x ! x ! FH(A)

--------------------------------------------------------------
2 ! Tonfrequenz Kanal B ! FL(B)
3 ! ! x ! x ! x ! x ! FH(B)

--------------------------------------------------------------
4 ! Tonfrequenz Kanal C ! FL(C)
5 ! ! x ! x ! x ! x ! FH(C)

--------------------------------------------------------------
6 ! Rauschfrequenz ! x ! x ! x ! RP

---------------------------------------------------------
7 ! Auswahl Ton- bzw. ! ! Rauschen ! Ton

! Rauschkanal ! x ! x ! C ! B ! A ! C ! B ! A
--------------------------------------------------------

8 ! A ! x ! x ! x ! H ! L(A)
9 ! Lautstärke Kanal B ! x ! x ! x ! H ! L(B)

10 ! C ! x ! x ! x ! H ! L(C)
--------------------------------------------------------------
11 ! Hüllkurvenfrequenz ! FHL
12 ! ! FHH
----------------------------------------------------
13 ! Hüllkurvenform ! x ! x ! x ! x ! HKF
------------------------------------------------------------



Parameter für Tonfrequenz

FW = 3750000 / (32 * F) ; F - reale Tonfrequenz (in Hz)
FH(n) = FW /256
FL(n) = FW mod 256

Rauschparameter

RP = 3750000 / (32 * F) ; F - max. Frequenzanteil im Rauschen
(3780 Hz < F < 117188 Hz)

Kanalauswahl

! 7 6 5 4 3 2 1 0 !
! x ! x ! ! ! ! ! ! ! Bit n = 0/1 --> Kanal aus/ein

---------------------------
! ! ! ! ! !
! ! ! ! ! ---- Kanal A Tonausgabe
! ! ! ! -------- " B "
! ! ! ------------ " C "
! ! !
! ! ---------------- " A Rauschen
! -------------------- " B "
------------------------ " C "

Lautstärkeeinstellung

! 7 6 5 4 3 2 1 0 !
! x ! x ! x ! 1 ! ! ! !

---------------------------
! ! ! ! !
! ---------------------- Lautstärke L (0...15)
!
-------------------------- =1 Aktivierung Hüllkurve

=0 konstante Lautstärke
entsprechend Bit 0-3

Hüllkurvenform (HKF)

Nr. ! Bit ! Kurvenform
! 7..4 ! 3 ! 2 ! 1 ! 0 !
------------------------

0 ! x ! 0 ! 0 ! x ! x ! @_____ x - beliebig (0/1)
4 ! x ! 0 ! 1 ! x ! x ! /_____
8 ! x ! 1 ! 0 ! 0 ! 0 ! @@@@@@
9 ! x ! 1 ! 0 ! 0 ! 1 ! @_____

10 ! x ! 1 ! 0 ! 1 ! 0 ! @/@/@/
11 ! x ! 1 ! 0 ! 1 ! 1 ! @
12 ! x ! 1 ! 1 ! 0 ! 0 ! //////
13 ! x ! 1 ! 1 ! 0 ! 1 ! /
14 ! x ! 1 ! 1 ! 1 ! 0 ! /@/@/@
15 ! x ! 1 ! 1 ! 1 ! 1 ! /_____

